Section 2. Projective Methods

Section 2.1. Galerkin Method

Section 2.1.1. Solving the initial problem.

First, let us solve the initial problem

```
In[29]:= Clear[y, x]
s = NDSolve[{y''[x] + x^3 y'[x] - (2 + Log[x + 1]^2) y[x] == 3, -y'[0] + 3 y[0] == 1, y[2] == 2}, y, {x, 0, 2}]
Plot[Evaluate[y[x] /. s], {x, 0, 2},
PlotRange → All,
PlotTheme → "Detailed",
AxesLabel → {"x", "y"},
GridLines → Automatic,
PlotStyle → Blue]
```

```
Out[30]=
{{y → InterpolatingFunction[ Domain: {{0., 2.}} Output: scalar ]}}
```

Out[31]=



Section 2.1.2. Choosing the basis functions

Now, choose basis functions. Choosing the 0th basis function is trivial:

In[175]:=

```
ϕ0[x_] = 5 x / 7 + 4 / 7;
(-ϕ0'[x] + 3 ϕ0[x]) /. {x → 0}(* Must be 1 *)
ϕ0[2] (* Must be 2 *)
```

Out[176]=

1

Out[177]=

2

Next functions will be chosen as the system of linearly independent polynomial of the special form:

In[178]:=

```
Clear[ϕn, x, an, bn, n]
ϕn[x_] = (x - 2) ^ n (an x + bn);
boundaryn[x_] = -ϕn'[x] + 3 ϕn[x]
Solve[boundaryn[0] == 0, {an, bn}]
```

Out[180]=

$-\text{an}\,(-2+x)^{n} - n\,(-2+x)^{-1+n}\,(\text{bn}+\text{an}\,x) + 3\,(-2+x)^{n}\,(\text{bn}+\text{an}\,x)$

Solve: Equations may not give solutions for all "solve" variables.

Out[181]=

$\left\{\left\{\text{bn} \to \frac{2\,\text{an}}{6+n}\right\}\right\}$

Thus, we are ready

```
n = 5; (* Number of basis functions. For n>5 I reach the limit of Wolfram Cloud *)
ϕ0[x_] = 5 x / 7 + 4 / 7;
ϕ = Table[(x - 2) ^ k ((6 + k) x + 2), {k, 1, n}];
Print["Basis Functions: ", ϕ0[x], " and ", ϕ]
```

Basis Functions: $\frac{4}{7} + \frac{5x}{7}$ and

$\{(-2+x)\,(2+7x),\ (-2+x)^{2}\,(2+8x),\ (-2+x)^{3}\,(2+9x),\ (-2+x)^{4}\,(2+10x),\ (-2+x)^{5}\,(2+11x)\}$

Verify that these basis functions satisfy boundary conditions as needed

In[206]:=

```
Table[(-D[ϕ[[k]], x] + 3 ϕ[[k]]) /. {x → 0}, {k, 1, n}] (* Must be all 0 *)
Table[ϕ[[k]] /. {x → 2}, {k, 1, n}] (* Must be all 0 as well *)
```

Out[206]=

{0, 0, 0, 0, 0}

Out[207]=

{0, 0, 0, 0, 0}

Setting the approximation:

In[228]:=

```
Clear[c]
coeffs = Array[Subscript[c, #] &, {n}];
u[x_] = ϕ0[x] + Sum[coeffs[[k]] × ϕ[[k]], {k, 1, n}]
```

Out[230]=

$$\frac{4}{7} + \frac{5\,x}{7} + (-2 + x)\,(2 + 7\,x)\,c_1 + (-2 + x)^2\,(2 + 8\,x)\,c_2 + (-2 + x)^3\,(2 + 9\,x)\,c_3 + (-2 + x)^4\,(2 + 10\,x)\,c_4 + (-2 + x)^5\,(2 + 11\,x)\,c_5$$

Section 2.1.3. Optimizing w.r.t. coefficients

In[232]:=

```
L[y_] = D[y[x], {x, 2}] + x^3 D[y[x], x] - (2 + Log[1 + x]^2) y[x];
R[x_] = L[u] - 3;
eqs = Table[Integrate[R[x] × ϕ[[k]], {x, 0.0, 2.0}] == 0, {k, 1, n}];
```

In[235]:=

```
Print[eqs]
```

$$\{67.9364 - 569.638\,c_1 + 697.672\,c_2 - 921.527\,c_3 + 1311.69\,c_4 - 1979.89\,c_5 == 0,$$
$$-80.3527 + 573.329\,c_1 - 939.299\,c_2 + 1487.47\,c_3 - 2402.02\,c_4 + 3978.31\,c_5 == 0,$$
$$110.273 - 731.356\,c_1 + 1399.7\,c_2 - 2486.13\,c_3 + 4369.65\,c_4 - 7717.86\,c_5 == 0,$$
$$-164.923 + 1053.25\,c_1 - 2222.04\,c_2 + 4267.69\,c_3 - 7968.23\,c_4 + 14\,756.3\,c_5 == 0,$$
$$260.807 - 1628.81\,c_1 + 3676.88\,c_2 - 7473.44\,c_3 + 14\,600.3\,c_4 - 28\,041.7\,c_5 == 0\}$$

In[237]:=

```
optimalCoeffs = Solve[eqs, coeffs]
```

Out[237]=

$$\{\{c_1 \to 0.0401299,\ c_2 \to -0.0516019,\ c_3 \to 0.0587259,\ c_4 \to 0.0514798,\ c_5 \to 0.0113561\}\}$$

In[243]:=

```
optimalCoeffs = Table[optimalCoeffs[[1]][[k]][[2]], {k, 1, n}]
```

Out[243]=

```
{0.0401299, -0.0516019, 0.0587259, 0.0514798, 0.0113561}
```

In[244]:=

```
uOptimal[x_] = ϕ0[x] + Sum[optimalCoeffs[[k]] × ϕ[[k]], {k, 1, n}]
```

Out[244]=

$$\frac{4}{7} + \frac{5\,x}{7} + 0.0401299\,(-2 + x)\,(2 + 7\,x) - 0.0516019\,(-2 + x)^2\,(2 + 8\,x) + 0.0587259\,(-2 + x)^3\,(2 + 9\,x) + 0.0514798\,(-2 + x)^4\,(2 + 10\,x) + 0.0113561\,(-2 + x)^5\,(2 + 11\,x)$$

In[246]:=

```
Plot[{Evaluate[y[x] /. s], uOptimal[x]}, {x, 0, 2},
PlotRange → All,
PlotTheme → "Detailed",
AxesLabel → {"x", "y"},
GridLines → Automatic,
PlotStyle → {Blue, Directive[Dashed, Red]}]
```

Out[246]=



Print absolute differences in the specified set of points:

In[299]:=

```
m = 20;
xs = Table[2 k / m, {k, 1, m}];
t = Table[Abs[uOptimal[xs[[k]]] - y[xs[[k]]] /. s], {k, 1, m}]
Max[t]
```

Out[301]=

```
{{0.000381356}, {0.000788737}, {0.00121535}, {0.000774551},
 {0.000114845}, {0.000905342}, {0.00119906}, {0.0008864}, {0.000151172},
 {0.000635991}, {0.00109295}, {0.00100677}, {0.000445847}, {0.000258035},
 {0.000675533}, {0.000539131}, {0.0000356648}, {0.000543849}, {0.00045365}, {0.}}
```

Out[302]=

```
0.00121535
```

Section 2.2. Least Squares Method

We simply change the system of equations for finding coefficients. The rest is the same.

In[304]:=

```
eqs = Table[Integrate[R[x] × D[R[x], coeffs[[k]]], {x, 0.0, 2.0}] == 0, {k, 1, n}];
Print[eqs]
```

$$\{-495.681 + 6912.42\, c_1 - 3371.77\, c_2 + 3773.29\, c_3 - 5309.95\, c_4 + 8429.41\, c_5 == 0,$$
$$466.082 - 3371.77\, c_1 + 5498.65\, c_2 - 8663.26\, c_3 + 14\,891.7\, c_4 - 26\,879.8\, c_5 == 0,$$
$$-615.633 + 3773.29\, c_1 - 8663.26\, c_2 + 17\,601.5\, c_3 - 35\,070.8\, c_4 + 69\,560.3\, c_5 == 0,$$
$$928.554 - 5309.95\, c_1 + 14\,891.7\, c_2 - 35\,070.8\, c_3 + 76\,975.6\, c_4 - 163\,204.\, c_5 == 0,$$
$$-1525.51 + 8429.41\, c_1 - 26\,879.8\, c_2 + 69\,560.3\, c_3 - 163\,204.\, c_4 + 363\,413.\, c_5 == 0\}$$

In[306]:=

```
optimalCoeffs = Solve[eqs, coeffs]
```

Out[306]=

$$\{\{c_1 \to 0.040226,\ c_2 \to -0.0523426,\ c_3 \to 0.0557251,\ c_4 \to 0.0486133,\ c_5 \to 0.0105585\}\}$$

In[307]:=

```
optimalCoeffs = Table[optimalCoeffs[[1]][[k]][[2]], {k, 1, n}]
uOptimal[x_] = ϕ0[x] + Sum[optimalCoeffs[[k]] × ϕ[[k]], {k, 1, n}]
```

Out[307]=

{0.040226, -0.0523426, 0.0557251, 0.0486133, 0.0105585}

Out[308]=

$$\frac{4}{7} + \frac{5\,x}{7} + 0.040226\,(-2 + x)\,(2 + 7\,x) - 0.0523426\,(-2 + x)^2\,(2 + 8\,x) +$$
$$0.0557251\,(-2 + x)^3\,(2 + 9\,x) + 0.0486133\,(-2 + x)^4\,(2 + 10\,x) + 0.0105585\,(-2 + x)^5\,(2 + 11\,x)$$

In[309]:=

```
Plot[{Evaluate[y[x] /. s], uOptimal[x]}, {x, 0, 2},
PlotRange → All,
PlotTheme → "Detailed",
AxesLabel → {"x", "y"},
GridLines → Automatic,
PlotStyle → {Blue, Directive[Dashed, Red]}]
```

Out[309]=



In[310]:=

```
m = 20;
xs = Table[2 k / m, {k, 1, m}];
t = Table[Abs[uOptimal[xs[[k]]] - y[xs[[k]]] /. s], {k, 1, m}]
Max[t]
```

Out[312]=

```
{{0.000476117}, {0.000991769}, {0.000728391}, {0.000186296},
 {0.0012365}, {0.00188564}, {0.00181404}, {0.00102341}, {0.000185937},
 {0.00134327}, {0.00199801}, {0.00190942}, {0.00116135}, {0.000140637},
 {0.000638744}, {0.000811352}, {0.000404109}, {0.000128277}, {0.000230791}, {0.}}
```

Out[313]=

```
0.00199801
```